Problem:
Score : 400 points

Problem Statement

Snuke has an empty sequence a.

He will perform N operations on this sequence.

In the i-th operation, he chooses an integer j satisfying 1 \leq j \leq i, and insert j at position j in a (the beginning is position 1).

You are given a sequence b of length N. Determine if it is possible that a is equal to b after N operations. If it is, show one possible sequence of operations that achieves it.

Constraints

All values in input are integers.

1 \leq N \leq 100

1 \leq b_i \leq N

Input

Input is given from Standard Input in the following format:

N
b_1 \dots b_N

Output

If there is no sequence of N operations after which a would be equal to b, print -1.
If there is, print N lines. In the i-th line, the integer chosen in the i-th operation should be printed. If there are multiple solutions, any of them is accepted.

Sample Input 1

3
1 2 1

Sample Output 1

1
1
2

In this sequence of operations, the sequence a changes as follows:

After the first operation: (1)

After the second operation: (1,1)

After the third operation: (1,2,1)

Sample Input 2

2
2 2

Sample Output 2

-1

2 cannot be inserted at the beginning of the sequence, so this is impossible.

Sample Input 3

9
1 1 1 2 2 1 2 3 2

Sample Output 3

1
2
2
3
1
2
2
1
1

Solution in Fortran (gfortran v4.8.4) (judge verdict: Accepted)

program main
  implicit none
  integer :: n, i
  integer, allocatable :: b(:), res(:)

  read (*, *) n
  allocate (b(n), res(n))
  read (*, *) b(:)

  do i = 1, n
    if (b(i) > i) then
      write (*, "(i0)") -1
      stop
    end if
  end do

  do i = 1, n
    call remove(b, res(n - i + 1))
  end do

  do i = 1, n
    write (*, "(i0)") res(i)
  end do

  contains

  subroutine remove(b, ret)
    integer, allocatable, intent(inout) :: b(:)
    integer, intent(out) :: ret
    integer :: n, i, c(size(b))
    n = size(b)
    c = [(b(i) - i, i = 1, n)]
    loop: block
      do i = n, 1, -1
        if (c(i) == 0) exit loop
      end do
      stop 1
    end block loop
    ret = b(i)
    b = [b(:i - 1), b(i + 1:)]
  end subroutine remove
end program main